

VLOOKUP function

Syntax

VLOOKUP (lookup_value, table_array, col_index_num, [range_lookup])

Argument name	Description
lookup_value (required)	<p>The value you want to look up. The value you want to look up must be in the first column of the range of cells you specify in the table_array argument.</p> <p>For example, if table_array spans cells B2:D7, then your lookup_value must be in column B.</p> <p>Lookup_value can be a value or a reference to a cell.</p>
table_array (required)	<p>The range of cells in which the VLOOKUP will search for the lookup_value and the return value. You can use a named range or a table, and you can use names in the argument instead of cell references.</p> <p>The first column in the cell range must contain the lookup_value. The cell range also needs to include the return value you want to find.</p> <p>Learn how to select ranges in a worksheet.</p>
col_index_num (required)	The column number (starting with 1 for the left-most column of table_array) that contains the return value.
range_lookup (optional)	<p>A logical value that specifies whether you want VLOOKUP to find an approximate or an exact match:</p> <ul style="list-style-type: none">▪ Approximate match - 1/TRUE assumes the first column in the table is sorted either numerically or alphabetically, and will then search for the closest value. This is the default method if you don't specify one. For example, =VLOOKUP(90,A1:B100,2,TRUE).▪ Exact match - 0/FALSE searches for the exact value in the first column. For example, =VLOOKUP("Smith",A1:B100,2,FALSE).

Examples

Here are a few examples of VLOOKUP:

Example 1

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=VLOOKUP(B3,B2:E7,2,FALSE)			
11	Result	Olivier			
12					

VLOOKUP looks for *Fontana* in the first column (column B) in table_array B2:E7, and returns *Olivier* from the second column (column C) of the table_array. FALSE returns an exact match.

Example 2

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=VLOOKUP(102,A2:C7,2,FALSE)			
11	Result	Fontana			

VLOOKUP looks for an exact match (FALSE) of the last name for 102 (lookup_value) in the second column (column B) in the A2:C7 range, and returns *Fontana*.

Example 3

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=IF(VLOOKUP(103,A1:E7,2,FALSE)="Sousa","Located","Not found")			
11	Result	Not found			

IF checks to see if VLOOKUP returns *Sousa* as the last name of employee corresponding to 103 (lookup_value) in A1:E7 (table_array). Because the last name corresponding to 103 is *Leal*, the IF condition is false, and *Not found* is displayed.

Example 4

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=INT(YEARFRAC(DATE(2014,6,30), VLOOKUP(105,A2:E7,5, FALSE), 1))			
11	Result	59			
12					
13					
14					
15					

VLOOKUP looks for the birth date of the employee corresponding to 105 (lookup_value) in the A2:E7 range (table_array), and returns 03/04/1955. Then, YEARFRAC subtracts this birth date from 2014/6/30 and returns a value, which is then converted by INT to the integer 59.

Example 5

	A	B	C	D	E
1	ID	Last name	First name	Title	Birth date
2	101	Davis	Sara	Sales Rep	12/08/68
3	102	Fontana	Olivier	VP (Sales)	02/19/52
4	103	Leal	Karina	Sales Rep	08/30/63
5	104	Patten	Michael	Sales Rep	09/19/58
6	105	Burke	Brian	Sales Manager	03/04/55
7	106	Sousa	Luis	Sales Rep	07/02/63
8					
9					
10	Formula	=IF(ISNA(VLOOKUP(105,A2:E7,2,FALSE)) = TRUE, "Employee not found", VLOOKUP(105,A2:E7,2,FALSE))			
11	Result	Burke	<p>IF checks to see if VLOOKUP returns a value for last name from column B for 105 (lookup_value). If VLOOKUP finds a last name, then IF will display the last name, otherwise IF returns <i>Employee not found</i>. ISNA makes sure that if VLOOKUP returns #N/A, then the error is replaced by <i>Employee not found</i>, instead of #N/A.</p> <p>In this example, the return value is <i>Burke</i>, which is the last name corresponding to 105.</p>		
12					
13					
14					

Common Problems

Problem	What went wrong
Wrong value returned	If range_lookup is TRUE or left out, the first column needs to be sorted alphabetically or numerically. If the first column isn't sorted, the return value might be something you don't expect. Either sort the first column, or use FALSE for an exact match.
#N/A in cell	<ul style="list-style-type: none">■ If range_lookup is TRUE, then if the value in the lookup_value is smaller than the smallest value in the first column of the table_array, you'll get the #N/A error value.■ If range_lookup is FALSE, the #N/A error value indicates that the exact number isn't found. <p>For more information on resolving #N/A errors in VLOOKUP, see How to correct a #N/A error in the VLOOKUP function.</p>
#REF! in cell	If col_index_num is greater than the number of columns in table_array , you'll get the #REF! error value. <p>For more information on resolving #REF! errors in VLOOKUP, see How to correct a #REF! error.</p>
#VALUE! in cell	If the table_array is less than 1, you'll get the #VALUE! error value. <p>For more information on resolving #VALUE! errors in VLOOKUP, see How to correct a #VALUE! error in the VLOOKUP function.</p>
#NAME? in cell	The #NAME? error value usually means that the formula is missing quotes. To look up a person's name, make sure you use quotes around the name in the formula. For example, enter the name as "Fontana" in =VLOOKUP("Fontana",B2:E7,2,FALSE). <p>For more information, see How to correct a #NAME! error.</p>
#SPILL! in cell	This particular #SPILL! error usually means that your formula is relying on implicit intersection for the lookup value, and using an entire column as a reference. For example, =VLOOKUP(A:A,A:C,2,FALSE). You can resolve the issue by anchoring the lookup reference with the @ operator like this: =VLOOKUP(@A:A,A:C,2,FALSE). Alternatively, you can use the traditional VLOOKUP method and refer to a single cell instead of an entire column: =VLOOKUP(A2,A:C,2,FALSE).

Best practices

Do this	Why
Use absolute references for <i>range_lookup</i>	<p>Using absolute references allows you to fill-down a formula so that it always looks at the same exact lookup range.</p> <p>Learn how to use absolute cell references.</p>
Don't store number or date values as text.	When searching number or date values, be sure the data in the first column of <i>table_array</i> isn't stored as text values. Otherwise, VLOOKUP might return an incorrect or unexpected value.
Sort the first column	Sort the first column of the <i>table_array</i> before using VLOOKUP when <i>range_lookup</i> is TRUE.
Use wildcard characters	<p>If <i>range_lookup</i> is FALSE and <i>lookup_value</i> is text, you can use the wildcard characters—the question mark (?) and asterisk (*)—in <i>lookup_value</i>. A question mark matches any single character. An asterisk matches any sequence of characters. If you want to find an actual question mark or asterisk, type a tilde (~) in front of the character.</p> <p>For example, =VLOOKUP("Fontan?",B2:E7,2,FALSE) will search for all instances of Fontana with a last letter that could vary.</p>
Make sure your data doesn't contain erroneous characters.	<p>When searching text values in the first column, make sure the data in the first column doesn't have leading spaces, trailing spaces, inconsistent use of straight (' or ") and curly (' or ") quotation marks, or nonprinting characters. In these cases, VLOOKUP might return an unexpected value.</p> <p>To get accurate results, try using the CLEAN function or the TRIM function to remove trailing spaces after table values in a cell.</p>