

XLOOKUP function

Syntax

The XLOOKUP function searches a range or an array, and then returns the item corresponding to the first match it finds. If no match exists, then XLOOKUP can return the closest (approximate) match.

=XLOOKUP(lookup_value, lookup_array, return_array, [if_not_found], [match_mode], [search_mode])

Argument	Description
lookup_value	The value to search for
Required*	*If omitted, XLOOKUP returns blank cells it finds in lookup_array .
lookup_array	The array or range to search
Required	
return_array	The array or range to return
Required	
[if_not_found]	Where a valid match is not found, return the [if_not_found] text you supply.
Optional	If a valid match is not found, and [if_not_found] is missing, #N/A is returned.
[match_mode]	Specify the match type:
Optional	0 - Exact match. If none found, return #N/A. This is the default. -1 - Exact match. If none found, return the next smaller item. 1 - Exact match. If none found, return the next larger item. 2 - A wildcard match where *, ?, and ~ have special meaning .
[search_mode]	Specify the search mode to use:
Optional	1 - Perform a search starting at the first item. This is the default. -1 - Perform a reverse search starting at the last item. 2 - Perform a binary search that relies on lookup_array being sorted in <i>ascending</i> order. If not sorted, invalid results will be returned. -2 - Perform a binary search that relies on lookup_array being sorted in <i>descending</i> order. If not sorted, invalid results will be returned.

Examples

Example 1 uses XLOOKUP to look up a country name in a range, and then return its telephone country code. It includes the **lookup_value** (cell F2), **lookup_array** (range B2:B11), and **return_array** (range D2:D11) arguments. It doesn't include the **match_mode** argument, as XLOOKUP produces an exact match by default.

G2		=XLOOKUP(F2,B2:B11,D2:D11)					
	A	B	C	D	E	F	G
1		Country	Abr	Prefix		What is the dial code?	
2		China	CN	+86		Brazil	+55
3		India	IN	+91			
4		United States	US	+1			
5		Indonesia	ID	+62			
6		Brazil	BR	+55			
7		Pakistan	PK	+92			
8		Nigeria	NG	+234			
9		Bangladesh	BD	+880			
10		Russia	RU	+7			
11		Mexico	MX	+52			

Note: XLOOKUP uses a lookup array and a return array, whereas VLOOKUP uses a single table array followed by a column index number. The equivalent VLOOKUP formula in this case would be:
=VLOOKUP(F2,B2:D11,3,FALSE)

Example 2 looks up employee information based on an employee ID number. Unlike VLOOKUP, XLOOKUP can return an array with multiple items, so a single formula can return both employee name and department from cells C5:D14.

C2		=XLOOKUP(B2,B5:B14,C5:D14)	
A	B	C	D
1	Emp ID	Employee Name	Department
2	8389	Dianne Pugh	Finance
3			
4	Emp ID	Employee Name	Department
5	4390	Ned Lanning	Marketing
6	8604	Margo Hendrix	Sales
7	8389	Dianne Pugh	Finance
8	4937	Earlene McCarty	Accounting
9	8299	Mia Arnold	Operations
10	2643	Jorge Fellows	Executive
11	5243	Rose Winters	Sales
12	9693	Carmela Hahn	Finance
13	1636	Delia Cochran	Accounting
14	6703	Marguerite Cervantes	Marketing

Example 3 adds an **if_not_found** argument to the preceding example.

C2* fx =XLOOKUP(B2,B5:B14,C5:D14,"ID not found")			
A	B	C	D
1	Emp ID	Employee Name	Department
2	1234	ID not found	
3			
4	Emp ID	Employee Name	Department
5	4390	Ned Lanning	Marketing
6	8604	Margo Hendrix	Sales
7	8389	Dianne Pugh	Finance
8	4937	Earlene McCarty	Accounting
9	8299	Mia Arnold	Operations
10	2643	Jorge Fellows	Executive
11	5243	Rose Winters	Sales
12	9693	Carmela Hahn	Finance
13	1636	Delia Cochran	Accounting
14	6703	Marguerite Cervantes	Marketing

Example 4 looks in column C for the personal income entered in cell E2, and finds a matching tax rate in column B. It sets the **if_not_found** argument to return 0 (zero) if nothing is found. The **match_mode** argument is set to 1, which means the function will look for an exact match, and if it can't find one, it returns the next larger item. Finally, the **search_mode** argument is set to 1, which means the function will search from the first item to the last.

F2* fx =XLOOKUP(E2,C2:C7,B2:B7,0,1,1)					
A	B	C	D	E	F
1	Tax Rate	Max Income		Income	Tax Rate
2	10%	\$9,700		\$46,523	24%
3	22%	\$39,475			
4	24%	\$84,200			
5	32%	\$160,726			
6	35%	\$204,100			
7	37%	\$510,300			

Note: XARRAY's **lookup_array** column is to the right of the **return_array** column, whereas VLOOKUP can only look from left-to-right.

Example 5 uses a nested XLOOKUP function to perform both a vertical and horizontal match. It first looks for **Gross Profit** in column B, then looks for **Qtr1** in the top row of the table (range C5:F5), and finally returns the value at the intersection of the two. This is similar to using the **INDEX** and **MATCH** functions together.

Tip: You can also use XLOOKUP to replace the **HLOOKUP** function.

Formula bar: `=XLOOKUP(D2,$B6:$B17,XLOOKUP($C3,$C5:$G5,$C6:$G17))`

	Quarter	Gross Profit	Net Profit	Profit %
	Qtr1	\$25,000	\$19,342	29.3%

Income Statement	Qtr1	Qtr2	Qtr3	Qtr4	Total
Total sales	\$50,000	\$78,200	\$89,500	\$91,250	\$308,950
Cost of sales	(\$25,000)	(\$42,050)	(\$59,450)	(\$60,450)	(\$186,950)
Gross profit	\$25,000	\$36,150	\$30,050	\$30,800	\$122,000
Depreciation	(\$899)	(\$791)	(\$202)	(\$412)	(\$2,304)
Interest	(\$513)	(\$853)	(\$150)	(\$956)	(\$2,472)
Earnings before Tax	\$23,588	\$34,506	\$29,698	\$29,432	\$117,224
Tax	(\$4,246)	(\$6,211)	(\$5,346)	(\$5,298)	(\$21,100)
Net profit	\$19,342	\$28,295	\$24,352	\$24,134	\$96,124
Profit %	29.3%	27.8%	23.4%	27.6%	26.9%

Note: The formula in cells D3:F3 is: `=XLOOKUP(D2,$B6:$B17,XLOOKUP($C3,$C5:$G5,$C6:$G17))`.

Example 6 uses the **SUM** function, and two nested XLOOKUP functions, to sum all the values between two ranges. In this case, we want to sum the values for grapes, bananas, and include pears, which are between the two.

Formula bar: `=SUM(XLOOKUP(B3,B6:B10,E6:E10):XLOOKUP(C3,B6:B10,E6:E10))`

Start	End	Total
Grape	Banana	\$110.70

Product	Qty	Price	Total
Apple	23	\$0.52	\$11.90
Grape	98	\$0.77	\$75.28
Pear	75	\$0.24	\$18.16
Banana	95	\$0.18	\$17.25
Cherry	42	\$0.16	\$6.80

The formula in cell E3 is: `=SUM(XLOOKUP(B3,B6:B10,E6:E10):XLOOKUP(C3,B6:B10,E6:E10))`

How does it work? XLOOKUP returns a range, so when it calculates, the formula ends up looking like this: `=SUM(E7:E9)`. You can see how this works on your own by selecting a cell with an XLOOKUP formula similar to this one, then select **Formulas > Formula Auditing > Evaluate Formula**, and then select **Evaluate** to step through the calculation.